

# Slowed down by LE? Perhaps the CEEPIPI service can help!



Thomas Petrolino IBM Poughkeepsie tapetro@us.ibm.com





#### **Trademarks**

The following are trademarks of the International Business Machines Corporation in the United States and/or other countries.

- •CICS®
- •DB2®
- ·Language Environment®
- •OS/390®
- •7/OS®

#### The following are trademarks or registered trademarks of other companies.

Java and all Java-related trademarks and logos are trademarks of Sun Microsystems, Inc., in the United States and other countries.

Linux is a registered trademark of Linus Torvalds in the United States, other countries, or both.

Microsoft, Windows and Windows NT are registered trademarks of Microsoft Corporation.

UNIX is a registered trademark of The Open Group in the United States and other countries.

SET and Secure Electronic Transaction are trademarks owned by SET Secure Electronic Transaction LLC.

\* All other products may be trademarks or registered trademarks of their respective companies.

#### Notes:

Performance is in Internal Throughput Rate (ITR) ratio based on measurements and projections using standard IBM benchmarks in a controlled environment. The actual throughput that any user will experience will vary depending upon considerations such as the amount of multiprogramming in the user's job stream, the I/O configuration, the storage configuration, and the workload processed. Therefore, no assurance can be given that an individual user will achieve throughput improvements equivalent to the performance ratios stated here.

IBM hardware products are manufactured from new parts, or new and serviceable used parts. Regardless, our warranty terms apply.

All customer examples cited or described in this presentation are presented as illustrations of the manner in which some customers have used IBM products and the results they may have achieved. Actual environmental costs and performance characteristics will vary depending on individual customer configurations and conditions.

This publication was produced in the United States. IBM may not offer the products, services or features discussed in this document in other countries, and the information may be subject to change without notice. Consult your local IBM business contact for information on the product or services available in your area.

All statements regarding IBM's future direction and intent are subject to change or withdrawal without notice, and represent goals and objectives only.

Information about non-IBM products is obtained from the manufacturers of those products or their published announcements. IBM has not tested those products and cannot confirm the performance, compatibility, or any other claims related to non-IBM products. Questions on the capabilities of non-IBM products should be addressed to the suppliers of those products.

Prices subject to change without notice. Contact your IBM representative or Business Partner for the most current pricing in your geography.

<sup>\*</sup> Registered trademarks of IBM Corporation




#### **Agenda**

- Understanding The Basics of PreInitialization
- Writing a Preinit Application
- Other Preinit Topics
- A Preinit Example
- Sources of Additional Information



## Understanding The Basics of Prelnitialization








#### **Background - LE Init/Term**

- Process Collection of Resources (LE message file, library code/data)
  - unaffected by HLL semantics, logically independent address space
- •Enclave Collection of Routines (Load modules, Heap, external data)
  - defines scope of HLL semantics, first routine is designated "main"
- •Thread "thread" of execution (Stack, raised conditions)
  - share the resources of the enclave





#### **Understanding The Basics**

- •Read <u>Language Environment Programming Guide</u>, Chapter 30 "Using preinitialization services" (SA22-7561)
- •Read <u>Language Environment Programming Guide for 64-bit</u>

  <u>Virtual Addressing Mode</u>, Chapter 22 "Using preinitialization services with AMODE 64" (SA22-7569)



#### **Understanding The Basics...**

- You can use preinitialization to enhance the performance of certain applications
- •Preinitialization lets a non-LE-conforming application (eg. Assembler) initialize an LE environment once, perform multiple executions of LE-conforming programs using that environment, and then explicitly terminate the LE environment
- •Because the environment is initialized only once (even if you perform multiple executions), you free up system resources and allow for faster responses to your requests.




#### A non-Preinit scenario

non-LE-conforming **LE-conforming** (LE not active) Initialize LE Run A Invoke A Terminate LE LE-conforming Initialize LE Run B Invoke B Terminate LE LE-conforming Initialize LE Run C Invoke C Terminate LE



#### Same application using Preinit

non-LE-conforming (LE not active)





#### Older forms of preinitialization

- •The following is a list of pre-LE language-specific forms of preinitialization. These environments are supported by LE but will not be enhanced.
  - C and PL/I -- supports prior form of C and PL/I preinitialization
     (PICI) through use of Extended Parameter List
  - ■C++ -- no prior form of preinitialization
  - COBOL -- supports the prior form of COBOL preinitialization through use of RTEREUS run-time option and ILBOSTP0 and IGZERRE functions
  - ■Fortran -- no prior form of preinitialization
- •LE Library Routine Retention (LRR) is also supported but is not the "preferred" method



#### Restrictions on pre-LE preinitialization

- •POSIX(ON)
- XPLINK
- AMODE 64




#### Users of preinitialization

- Numerous IBM products currently utilize preinitialization
  - Program Management Binder for C++ demangler
  - •DB2 for stored procedures
  - •CICS TS V3.I for XPLINK support
  - •...
- Many IBM customers...



### Writing a Preinit Application









- A Preinit application consists of:
  - One or more HLL routines
  - •A Preinit Table
  - A Preinit Assembler Driver



#### **HLL Routines**

#### An example subroutine: Notice anything unusual?

```
CBL LIB,QUOTE

IDENTIFICATION DIVISION.

PROGRAM-ID. HLLPIPI.

DATA DIVISION.

WORKING-STORAGE SECTION.

PROCEDURE DIVISION.

DISPLAY "COBOL subprogram beginning".

DISPLAY "Called using LE Preinitialization".

DISPLAY "Call subroutine interface.".

DISPLAY "COBOL subprogram returns to caller.".

GOBACK.
```

#### Your answer should be "Nope!"



#### **HLL Routines**

- Written in

  - **C**++
  - PL/I
  - COBOL
- May be main or subroutine
  - •If using an XPLINK or AMODE 64 subroutine, it must be declared "fetchable"



#### The Preinit table

- •The Preinit table identifies routines to be executed (and optionally loaded) in a Preinit environment
  - It contains routine names and/or entry point addresses
  - It is possible to have an "empty" Preinit table with empty rows
    - •routines can be added later using the Preinit add\_entry interface
- •In the Preinit table, entry point addresses are maintained with the High Order Bit set to indicate AMODE of routine
  - ■HOB on, routine is AMODE31 and invoked in 31 bit mode
  - ■HOB off, routine is AMODE24 and invoked in 24 bit mode
- •CEEBXITA (Asm User Exit), CEEBINT (HLL User Exit), CEEUOPT are obtained from first entry in Preinit table



#### **Generate the Preinit table**

- LE provides the following assembler macros to generate the Preinit table
  - **CEEXPIT** generates a header for the Preinit table
  - **CEEXPITY** generates an entry within the Preinit table
    - specify entry name and/or entry\_point address of the routine
    - each invocation generates a row in the Preinit table
    - •if name is blank and entry\_point is zero, then an empty row is added to the Preinit table
  - **CEEXPITS** identifies the end of the Preinit table
  - **•CELQPIT, CELQPITY, CELQPITS** for AMODE 64
- The size of the Preinit table cannot be increased dynamically



#### **The Preinit Table**

Declared in the data section of the Preinit Assembler Driver:

```
Preinitialization Table.
                                       Preinitialization Table with index
         CEEXPIT ,
PPTBL
         CEEXPITY HLLPIPI, 0
                                       dynamically loaded routine
                                       statically-bound routine
         CEEXPITY , HLLEXTRN
                                       empty Table slot
         CEEXPITY ,
                                       Endof PreInit table
         CEEXPITS ,
         EXTRN
                  HLLEXTRN
```



#### The Preinit Assembler Driver

- •The Preinit Assembler Driver is responsible for:
  - Loading the Preinit Interface module
  - Initializing / Terminating the Preinit environment
  - Calling HLL routines using the Preinit environment

## SHARE.ORG

#### The Preinit Interface Module

- •The main Preinit interface is the loadable module "CEEPIPI"
  - The AMODE 64 Preinit interface is the loadable module "CELQPIPI"
- •CEEPIPI handles the requests and provides services for:
  - **LE Environment Initialization**
  - Application Invocation
  - **LE Environment Termination**
- •All requests for services by CEEPIPI must be made from a non-Language Environment environment
- The parameter list for CEEPIPI is an OS standard linkage parameter list
  - •First parameter on each call to CEEPIPI is a Preinit function code





\*

\* Load LE CEEPIPI service routine into main storage.

\*

LOAD EP=CEEPIPI
ST R0, PPRTNPTR

Load CEEPIPI routine dynamically

Save the addr of CEEPIPI routine

\*

:



#### **Preinit Initialization**

- •LE supports three forms of preinitialized environments
- They are distinguished by the level of initialization
  - •init\_main supports the execution of main routine
    - •initializes LE environment through process-level
    - each call\_main invocation initializes enclave- and thread-level
  - •init\_sub supports the execution of subroutines
    - •initializes LE environment through process-, enclave-, and thread-level
    - •each call\_sub invocation has minimal overhead
  - •init\_sub\_dp a special form of the init\_sub that allows multiple preinitialized environments, for executing subroutines, to be created under the same task (TCB). For AMODE 64 init\_sub is comparable.
    - Only one POSIX(ON) environment per TCB



#### **Preinit Initialization...**

#### •main Environment

- Advantages
  - •A new, pristine environment is created
  - •Run-Time options can be specified for each application
- Disadvantages
  - Poorer performance

#### •sub Environment

- Advantages
  - Best performance
- Disadvantages
  - •The environment is left in what ever state the previous application left it (including WSA, working storage, etc)
  - Run-Time options cannot be changed



#### Initializing a Preinit Environment

\* \* Initialize an LE Preinitialization main environment. \* INIT ENV EQU R5, PPTBL Get address of Preinit Table LA ST R5, @CEXPTBL Ceexptbl addr ->Preinit Table R15, PPRTNPTR Get address of CEEPIPI routine L \* Invoke CEEPIPI routine (15), (INITMAIN, @CEXPTBL, @SRVRTNS, TOKEN) CALL \* Check return code: R2, R15 Is R15 = zero? LTR Yes (success)..go to next section BZCMAIN \* No (failure)..issue message WTO 'ASMPIPI: call to (INIT MAIN) failed', ROUTCDE=11 R2, =F'8'Check for partial initialization Yes..go do Preinit termination BE TMAIN \* No..issue message & quit 'ASMPIPI: INIT MAIN failure RC is not 8.', ROUTCDE=11 WTO ABEND (R2), DUMP Abend with bad RC and dump memory



#### Initializing a Preinit Environment

SHARE. ORG

```
*
* Initialize an LE Preinitialization subroutine environment.
*
INIT ENV EQU
            R5, PPTBL
                                        Get address of Preinit Table
         LA
         ST R5, @CEXPTBL
                                       Ceexptbl addr ->Preinit Table
               R15, PPRTNPTR
                                       Get address of CEEPIPI routine
         L
* Invoke CEEPIPI routine
              (15), (INITSUB, @CEXPTBL, @SRVRTNS, RUNTMOPT, TOKEN)
         CALL
* Check return code:
               R2, R15
                                        Is R15 = zero?
         LTR
              CSUB
                                        Yes (success)..go to next section
         BZ
* No (failure)..issue message
         WTO 'ASMPIPI: call to (INIT SUB) failed', ROUTCDE=11
               R2, =F'8'
                                        Check for partial initialization
                                        Yes..go do Preinit termination
         BE
               TSUB
* No..issue message & quit
               'ASMPIPI: INIT SUB failure RC is not 8.', ROUTCDE=11
         WTO
         ABEND (R2), DUMP
                                        Abend with bad RC and dump memory
```



#### Calling the HLL Routine

- •Language Environment provides services to invoke either a main routine or subroutine.
  - •When invoking main routines, the environment must have been initialized with init\_main
  - •When invoking subroutines, the environment must have been initialized with init\_sub or init\_sub\_dp
- •The Preinit environment identified by **token** is activated before the specified routine is called
- After the called routine returns, the environment becomes "dormant"
- •The parameter list is passed to the application as-is
  - XPLink & 64-bit convert from OS format to XPLink



#### Calling the HLL Routine...

SHARE. ORG

- •It is important to provide the parameter list in the exact format that the compiled routine is expecting
  - C Example: 'TESTPGM 10 5' when interactively invoked
  - C function prototype: main(int argc, char \*\*argv)
  - Assembler parameter list layout:

PARMPTR *	DC A(	PARMLIST)	Pointer to PARMLIST
PARMLIST	DS OA		Parameter List
ARGC	DC F'	3'	Number of arguments
ARGVPTR	DC A(	ARGV)	Pointer to Argument Array
*			
ARGV	DS	0 <b>A</b>	Argument Array
ARCV0	DC	A (ARGVOS)	Pointer to Argument 1
ARGV1	DC	A (ARGV1S)	Pointer to Argument 2
ARGV2	DC	A (ARGV2S)	Pointer to Argument 3
*			
ARGV0S	DC	C'TESTPGM', X'00'	Argument 1
ARGV1S	DC	C'10', X'00'	Argument 2
ARGV2S	DC	C'5', X'00'	Argument 3



#### Calling a HLL Main

\* \* Call the main, which is loaded by LE CMAIN EOU L R15, PPRTNPTR Get address of CEEPIPI routine CALL (15), (CALLMAIN, PTBINDEX, TOKEN, RUNTMOPT, PARMPTR, X ENCRETC, ENCRSNC, APPLFBC) \* Check return code: Is R15 = zero? LTR R2, R15 **TMATN** BZYes (success)..go to next section No (failure) .. issue message & quit 'ASMPIPI: call to (CALL\_MAIN) failed', ROUTCDE=11 WTO ABEND (R2), DUMP Abend with bad RC and dump memory



#### Calling a HLL Subroutine

\* \* Call the subroutine, which is loaded by LE **CSUB** EOU Get address of CEEPIPI routine L R15, PPRTNPTR CALL (15), (CALLSUB, PTBINDEX, TOKEN, PARMPTR, X SUBRETC, SUBRSNC, SUBFBC) \* Check return code: Is R15 = zero? R2, R15 LTR TSUB Yes (success)..go to next section BZ. No (failure) .. issue message & quit WTO 'ASMPIPI: call to (CALL\_SUB) failed', ROUTCDE=11 ABEND (R2), DUMP Abend with bad RC and dump memory



#### **Preinit Termination**

- The Preinit application terminates the Preinit environment once it is no longer needed
- Termination performs cleanup of the resources associated with the environment
- A single Termination service handles all types of Preinit environments





#### **Terminating the Preinit Environment**

SHARE. ORG

```
*
* Terminate the environment
         EQU
TSUB
               R15, PPRTNPTR
                                      Get address of CEEPIPI routine
              (15), (TERM, TOKEN, ENV RC)
* Check return code:
         LTR
               R2, R15
                                       Is R15 = zero?
         BZ
               DONE
                                      Yes (success)..go to next section
* No (failure)..issue message & quit
         WTO
               'ASMPIPI: call to (TERM) failed', ROUTCDE=11
         ABEND (R2), DUMP
                                      Abend with bad RC and dump memory
```

#### **Other Preinit Topics**







#### **Reentrancy Considerations**

- You can make multiple calls to main routines or subroutines
- In general, you should specify only reentrant routines for multiple invocations:
  - •Multiple calls to a reentrant main routine are not influenced by a previous execution of the same routine
  - •For example, external variables are reinitialized for every call to a reentrant **main**
  - If you have a nonreentrant COBOL program, condition IGZ0044S is signalled when the routine is invoked again
  - If you have a nonreentrant C main() program that uses external variables, then when your routine is invoked again, the variables will be in last-use state
  - Multiple calls to reentrant **subroutines** reuse the same working storage, it is only initialized once during (*call\_sub*)



#### **Stop Semantics in Preinit subs**

- •When one of the following occurs within a preinitialized environment for subroutines, the logical enclave is terminated:
  - •C exit(), abort(), or signal handling function specifying a normal or abnormal termination
  - COBOL STOP RUN statement
  - ■PL/I STOP or EXIT
  - an unhandled condition causing termination of the (only) thread
- The process level of the environment is retained
- Modules in Preinit table are not deleted
- •The next call to a subrtn in this environment will initialize a new enclave (possibly with different user exits)



#### **Additional Preinit Services**

- Calling a Subroutine By Address
  - call\_sub\_addr: Invoke a subroutine by address within an already initialized environment
- Improving Performance of a Sequence of Calls
  - start\_seq: Start a sequence of uninterruptible calls to a number of subroutines
  - end\_seq: Terminate a sequence of uninterruptible calls to a number of subroutines



#### Additional Preinit Services...

- Managing the Preinit Table
  - add\_entry: Dynamically add a routine to an environment
  - delete\_entry: Delete an entry from the Preinit table,
     making it available to a later add\_entry
- Extracting Information from an Environment
  - identify\_environment: Determine characteristics of a Preinit environment
  - identify\_entry: Identify the language of an entry in the Preinit table
  - identify\_attributes: Identify the attributes of an entry in the Preinit table

# S H A R E Technology · Connections · Results

#### **User Exit Invocation**

SHARE. ORG

	init_sub, init_sub_dp	call_main	call_sub or call_sub_addr ended with STOP semantics	term for "clean" init_sub or init_sub_dp environment	term
CEEBXITA (enclave init)	X	X	X(next call)		
CEEBINT (HLL exit)	X	X	X(next call)		
C atexit() functions		×	×	×	
CEEBXITA (enclave term)		X	X	X	
CEEBXITA (process term)				X	X

- Main environments: CEEBXITA and CEEBINT application-specific user exits are taken from the main routine being called.
- Sub environments: CEEBXITA and CEEBINT application-specific user exits are taken from the first entry in Preinit table.
- All other occurrences are ignored!





- Preinit applications can run XPLINK-compiled programs in a Preinit environment.
- •LE initializes either an XPLINK environment or a "regular" (non-XPLINK) environment
  - Main: XPLINK environment if routine in first Preinit
     Table entry is XPLINK
  - Subroutine: XPLINK environment if routine in first Preinit Table entry is XPLINK, or if XPLINK(ON) run-time option is specified



## **XPLINK Preinit...**

- •call\_main may cause an environment switch
  - If running a non-XPLINK environment, and either the program was compiled XPLINK or XPLINK(ON) was specified, the environment will be rebuilt XPLINK, and remain that way.
- Sub environments do not switch
  - •A call to an XPLINK subroutine in a non-XPLINK environment will result in a "mismatch" error.
- Recommendation: Do not use non-XPLINK routines in an XPLINK Preinit environment.



### **Service routines**

- •Under Preinit, you can specify several service routines for use with running a main routine or subroutine in the preinitialized environment
- •To use the routines, specify a list of addresses of the routines in a service routine vector
  - ■Pass the address of this list on the init\_main, init\_sub, or init\_sub\_dp interfaces
  - ■The service\_rtns parameter that you specify contains the address of the vector itself
  - •If this pointer is specified as zero (0), LE routines are used instead of the service routines
- •Why?
  - Execution environment has its own storage or program management services
- Now supported in AMODE 64 Language Environment
  - z/OS VI.9: @Load and @Delete service routines
  - z/OS VI.II: @Getstore, @Freestore, and @Msgrtn service routines

# S H A R E Technology · Connections · Results

### Service routines...

- Count
  - the number of fullwords that follow
- User Word
  - passed to the service routines
  - •provides a means for your routine to communicate to the service routines
- •@Workarea
  - •address of a work area of at least 256 bytes that is doubled word aligned. First word contains the length of area provided. Required if service routines present in vector
- •@Load
  - •loads named routines for application management
- •@Delete
  - deletes routines for application management



### Service routines...

- •@Getstore
  - •allocates storage on behalf of the storage manager. This routine relies on the caller to provide a save area, which can be the @Workarea
- •@Freestore
  - •frees storage on behalf of storage manager
- •@Exceprtn
  - traps program interrupts and abends for condition management
- •@Msgrtn
  - allows error messages to be processed by caller of the application



- Preinit Trace Table
- IPCS Support to format Preinit control blocks and trace table



- Preinit Trace Table Characteristics
  - Tracing is always active
    - •Begins when the Preinit environment is initialized and ends when the environment is terminated
  - Trace is kept in an in-storage trace table
    - Fixed size (4096 bytes)
    - ·Wraps when the end has been reached



- •New keyword for the LEDATA IPCS Verbexit:
  - •PTBL(value) Formats Preinit control block and trace table based on value:
    - •"CURRENT" Preinit data associated with the current or specified TCB is displayed.
    - <address> Preinit data at that address is displayed.
    - ·"\*" Data for all active and dormant Preinit environments within the current address space are displayed; \*\*\* This option is time-consuming \*\*\*.
    - •"ACTIVE" Display Preinit data associated with each TCB in the address space.



#### LEDATA PTBL Output – Preinit Control Block

```
=== > VERBEXIT LEDATA 'PTBL (CURRENT)'
PreInitialization Programming Interface Trace Data
 CEEPIPI Environment Table Entry and Trace Entry:
 Active CEEPIPI Environment ( Address 25805CB0 )
  Evecatcher : CEEXIPTB
  TCB address: 008D1B08
 CEEPIPI Environment :
  Non-XPLINK Environment
  Environment Type : MAIN
  Sequence of Calls not active
  Exits not established
  Signal Interrupt Routines not registered
  Service Routines are not active
  CEEPIPI Environment Enclave Initialized
  Number of CEEPIPI Table Entries = 2
```



#### LEDATA PTBL Output – Preinit Control Block...

```
CEEPIPI Table Entry Information :

CEEPIPI Table Index 0 (Entry 1 )

Routine Name = HLLCRTN

Routine Type = C/C++

Routine Entry Point = A5810B38

Routine Function Pointer = A5810CC0

Routine Entry is Non-XPLINK

Routine was loaded by Language Environment

Routine Address was resolved

Routine Function Descriptor was valid

Routine Return Code = 0

Routine Reason Code = 0
```



#### LEDATA PTBL Output – Preinit Control Block...



#### LEDATA PTBL Output – Preinit Trace Table

```
CEEPIPI Trace Table Entries :

Call Type = INIT_MAIN

PIPI Driver Address = A5800A82

Load Service Return Code = 0

Load Service Reason Code = 0

Most Recent Return Code = 0

Most Recent Reason Code = 0

An ABEND will be issued if storage can not be obtained PreInit Environment will not allow EXEC CICS commands

Service RC = 0 :A new environment was initialized
```



#### LEDATA PTBL Output – Preinit Trace Table...

```
Call Type = ADD_ENTRY
Routine Table Index
                        = 1
Routine Name = HLLPIPI
Routine Address = A5812E20
Load Service Return Code
                        = 0
Load Service Reason Code = 3
Service RC = 0 : The routine was added to the PreInit table.
Call Type = CALL_MAIN
Routine Table Index = 1
Enclave Return Code = 0
Enclave Reason Code = 0
Service RC = 0 : The environment was activated and the
   routine called.
```




#### LEDATA PTBL Output – Preinit Trace Table...



The following example provides an illustration of an assembler program ASMPIPI ASSEMBLE invoking CEEPIPI to:

- •Initialize a LE Preinit subroutine environment
- •Load and call a reentrant C/COBOL/PLI subroutine
- •Terminate the LE Preinit environment







## **Example**

- •Following the assembler program are interchangeable examples of the program HLLPIPI written in:
  - ■C, COBOL, and PL/I
- •HLLPIPI is called by an assembler program, ASMPIPI.
- ASMPIPI uses the Language Environment preinitialized program subroutine call interface
- You can use the assembler program to call the HLL versions of HLLPIPI.



\*



```
* Standard program entry conventions.
ASMPTPT
         CSECT
               R14, R12, 12 (R13)
                                         Save caller's registers
         STM
         LR
               R12, R15
                                         Get base address
         USING ASMPIPI, R12
                                         Identify base register
         ST
               R13, SAVE+4
                                        Back-chain the save area
                                         Get addr of this routine's save area
         LA R15, SAVE
            R15,8(R13)
                                        Forward-chain in caller's save area
         ST
               R13, R15
                                        R13 -> save area of this routine
         LR
*
* Load LE CEEPIPI service routine into main storage.
*
         LOAD
               EP=CEEPIPI
                                         Load CEEPIPI routine dynamically
                                         Save the addr of CEEPIPI routine
         ST
               R0, PPRTNPTR
```



ABEND (R2), DUMP

\* \* Initialize an LE Preinitialization subroutine environment. \* INIT ENV EQU LA R5, PPTBL Get address of Preinit Table ST R5, @CEXPTBL Ceexptbl addr ->Preinit Table R15, PPRTNPTR Get address of CEEPIPI routine L \* Invoke CEEPIPI routine CALL (15), (INITSUB, @CEXPTBL, @SRVRTNS, RUNTMOPT, TOKEN) \* Check return code: R2, R15 Is R15 = zero? LTR CSUB Yes (success)..go to next section BZ\* No (failure)..issue message WTO 'ASMPIPI: call to (INIT SUB) failed', ROUTCDE=11 R2, =F'8'Check for partial initialization **TSUB** Yes..go do Preinit termination BE \* No..issue message & quit 'ASMPIPI: INIT SUB failure RC is not 8.', ROUTCDE=11 WTO

Abend with bad RC and dump memory



```
*
* Call the subroutine, which is loaded by LE
*
CSUB
         EQU
                                         Get address of CEEPIPI routine
         L
               R15, PPRTNPTR
         CALL (15), (CALLSUB, PTBINDEX, TOKEN, PARMPTR,
                                                                            X
                 SUBRETC, SUBRSNC, SUBFBC)
* Check return code:
         LTR
               R2, R15
                                         Is R15 = zero?
         BZ
               TSUB
                                         Yes (success)..go to next section
* No (failure)..issue message & quit
               'ASMPIPI: call to (CALL SUB) failed', ROUTCDE=11
         WTO
         ABEND (R2), DUMP
                                         Abend with bad RC and dump memory
```



LM

BR

\* \* Terminate the environment \* TSUB EQU L R15, PPRTNPTR Get address of CEEPIPI routine CALL (15), (TERM, TOKEN, ENV RC) \* Check return code: LTR R2,R15 Is R15 = zero? BZDONE Yes (success)..go to next section \* No (failure)..issue message & quit WTO 'ASMPIPI: call to (TERM) failed', ROUTCDE=11 ABEND (R2), DUMP Abend with bad RC and dump memory \* Standard exit code. DONE EQU R15,0 LA Passed return code for system R13, SAVE+4 Get address of caller's save area L L R14, 12 (R13) Reload caller's register 14

Reload caller's registers 0-12

Branch back to caller

R0, R12, 20 (R13)

R14



SHARE. ORG

```
* CONSTANTS and SAVE AREA.
                       18F'0'
SAVE
                DC
                DS
                                         Save the address of CEEPIPI routine
PPRTNPTR
                      Α
* Parameters passed to an (INIT SUB) call.
                      F'3'
                                         Function code to initialize for subr
INITSUB
                DC
                      A (PPTBL)
                                         Address of Preinitialization Table
@CEXPTBL
                DC
@SRVRTNS
                      A(0)
                DC
                                         Addr of service-rtns vector, 0 = none
                                         Fixed length string of runtime optns
                      CL255''
RUNTMOPT
                DC
                                         Unique value returned(output)
TOKEN
                DS
                      Ŧ
* Parameters passed to a (CALL_SUB) call.
                      F'4'
                                         Function code to call subroutine
CALLSUB
                DC
                      F'0'
PTBINDEX
                DC
                                         The row number of Preinit Table entry
PARMPTR
                DC
                      A(0)
                                         Pointer to @PARMLIST or zero if none
SUBRETC
                DS
                      F
                                         Subroutine return code (output)
SUBRSNC
                DS
                      F
                                         Subroutine reason code (output)
                DS
                       3F
                                          Subroutine feedback token (output)
SUBFBC
```



SHARE. ORG

## Example...

\* \* Parameters passed to a (TERM) call. DC F'5' Function code to terminate TERM ENV RC Environment return code (output) DS F \* Preinitialization Table. Preinitialization Table with index PPTBL CEEXPIT , CEEXPITY HLLPIPI, 0 0=dynamically loaded routine Endof PreInit table CEEXPITS , \* LTORG R0 EQU 0 EQU R1 1 R14 14 EQU R15 EQU 15 END ASMPIPI



#### C Subroutine Called by ASMPIPI

```
#include <stdio.h>

HLLPIPI ()
{
    printf("C subroutine beginning \n");
    printf("Called using LE PreInit call \n");
    printf("Subroutine interface.\n");
    printf("C subroutine returns to caller \n");
}
```



#### COBOL Program Called by ASMPIPI

```
CBL LIB, QUOTE
     *Module/File Name: IGZTPIPI
     ****************
     * HLLPIPI is called by an assembler program, ASMPIPI.
     * ASMPIPI uses the LE preinitialized program
     * subroutine call interface. HLLPIPI can be written
     * in COBOL, C, or PL/I.
     IDENTIFICATION DIVISION.
     PROGRAM-ID, HIJPIPI.
     DATA DIVISION.
     WORKING-STORAGE SECTION.
     PROCEDURE DIVISION.
         DISPLAY "COBOL subprogram beginning".
         DISPLAY "Called using LE Preinitialization".
         DISPLAY "Call subroutine interface.".
         DISPLAY "COBOL subprogram returns to caller.".
         GOBACK.
```



#### PL/I Routine Called by ASMPIPI

```
/*Module/File Name: IBMPIPI
                                                 * /
* /
/* HLLPIPI is called by an assembler program, ASMPIPI.
                                                 * /
/* ASMPIPI uses the LE preinitializedprogram
                                                 * /
/* subroutine call interface. HLLPIPI can be written
                                                 * /
/* in COBOL, C, or PL/I.
                                                 * /
                                                 * /
HLLPIPI: PROC OPTIONS (FETCHABLE);
        DCL RESULT FIXED BIN(31,0)INIT(0);
        PUT SKIP LIST
            ('HLLPIPI: PLI subroutine beginning.');
        PUT SKIP LIST
            ('HLLPIPI: CalledLE Preinit Call');
        PUT SKIP LIST
            ('HLLPIPI: Subroutine interface.');
        PUT SKIP LIST
            ('HLLPIPI: PLI program returns to caller.');
        RETURN;
END HLLPIPI;
```

SHARE in Anaheim, March 2011 - Copyright IBM Corp. 2001, 2011

## **Sources of Additional Information**



- •LE Debug Guide and Runtime Messages
- •LE Programming Reference
- LE Programming Guide (64-bit too!)
- LE Customization
- •LE Migration Guide
- LE Writing ILC Applications
- Web site
  - http://www.ibm.com/servers/eserver/zseries/zos/le/